

```

//*****
//      MACRO: LETTER.WCM
//      PURPOSE: Addresses and formats a standard business letter.
//*****
Application (A1; "WordPerfect"; Default; "UK")
OnError(Error@)
TemplateFill()
FillAgain:=False
If(?BlockActive<>0)
    SelectMode(Off!)
EndIf
While(?Substructure)
    SubstructureExit()
EndWhile
GroupName := "WordPerfect"
SubjectPrefix:="RE: "
Subject:=""
DefaultUnits(WPUnits!)

//*****
//      MAIN PROGRAM BEGIN
//*****
Call(GetSettings@)
Call(DefineMainDlg@)
Call(DisplayMainDlg@)
Call(SaveSettings@)
Call(DoLetter@)
Go(Stop@)
//*****
//      MAIN PROGRAM END
//*****

//*****
//      ROUTINE NAME: GetSettings@
//      INPUT VARIABLES: None
//      OUTPUT VARIABLES: HeaderOn; SubjectOn; ReturnOn; Author; Title; Format;
LetterType
//      USES: Nothing
//      DESCRIPTION: Retrieves letter settings from private BIF file.
//*****
Label(GetSettings@)
// Get Subject On/Off
BIFRead(Status: BIFStat; Group: GroupName; Section: "Letters"; Item: "Print Subject"; Value:
SubjectOn; ExpectedItemType: Boolean!; NameListType: Private!)
If(BIFStat<0)
    SubjectOn:=False

```

```

EndIf
// Get Return Address On/Off
BIFRead(Status: BIFStat; Group: GroupName; Section: "Letters"; Item: "Print Return Address";
Value: ReturnOn; ExpectedItemType: Boolean!; NameListType: Private!)
If(BIFStat<0)
    ReturnOn:=False
EndIf
// Get Header On/Off
BIFRead(Status: BIFStat; Group: GroupName; Section: "Letters"; Item: "Header for Second
Page"; Value: HeaderOn; ExpectedItemType: Boolean!; NameListType: Private!)
If(BIFStat<0)
    HeaderOn:=True
EndIf
// Get author name and title
BIFRead(Status: BIFStat; Group: GroupName; Section: "Template Information"; Item: "Personal
Information"; Value: PersInfo; ExpectedItemType: WPWordString!; NameListType: Private!)
If(BIFStat>-1)
    Author:=SubStr(PersInfo;1;StrPos(PersInfo;"|")-1)
    PersInfo:=SubStr(PersInfo;StrPos(PersInfo;"|")+1;StrLen(PersInfo))
    Title:=SubStr(PersInfo;1;StrPos(PersInfo;"|")-1)
    If(Author="")
        Author:="<Author Name>"
    EndIf
    If(Title="")
        Title:="<Author Title>"
    EndIf
Else
    Author:="<Author Name>"
    Title:="<Author Title>"
EndIf
// Get letter format
BIFRead(Status:BIFStat; Group:GroupName; Section: "Letters"; Item: "Format"; Value: Format;
ExpectedItemType: WPWordString!; NameListType: Private!)
If(BIFStat<0)
    Format:="Full Block"
EndIf
Return
//*****

//*****
//    ROUTINE NAME: DefineMainDlg@
//    INPUT VARIABLES: None
//    OUTPUT VARIABLES: None
//    USES: Nothing
//    DESCRIPTION: Defines main dialog.
//*****

```

Label(DefineMainDlg@)  
HRT:=NToC(63754)  
FullBlock:=

HRT+HRT+HRT+"Date"+HRT+HRT+"Address"+HRT+"Address"+HRT+HRT+"Salutation:"+  
HRT+HRT+"Lorem ipsum dolor sit amet. Thun amis ozz weepay yusil eepeet zaboy. Istee  
mundus feeri obundus."+HRT+HRT+"Lorem ipsum dolor sit amet istee mundus feeri  
obundus."+HRT+HRT+ "Closing,"+HRT+HRT+HRT+Author+HRT+Title

ModBlock:= HRT+HRT+HRT+"  
Date"+HRT+HRT+"Address"+HRT+ "Address"+HRT+HRT+"Salutation:"+HRT+HRT+"Lorem  
ipsum dolor sit amet. Thun amis ozz weepay yusil eepeet zaboy. Istee mundus feeri obundus." +  
HRT + HRT+"Lorem ipsum dolor sit amet istee mundus feeri obundus."+HRT+HRT+ "  
Closing,"+HRT+HRT+HRT+" "+Author+HRT+"  
"+Title

SemiBlock:= HRT+HRT+HRT+" Date"+HRT+HRT+"Address"+HRT  
+"Address"+HRT+HRT+"Salutation:"+HRT+HRT+" Lorem ipsum dolor sit amet. Thun  
amis ozz weepay yusil eepeet zaboy. Istee mundus feeri obundus."+HRT+HRT+" Lorem  
ipsum dolor sit amet istee mundus feeri obundus."+HRT+HRT+ "  
Closing,"+HRT+HRT+ HRT+" "+Author+HRT+"  
"+Title

Simplified:=

HRT+HRT+HRT+HRT+HRT+"Date"+HRT+HRT+"Address"+HRT+"Address"+HRT+HRT+  
"Lorem ipsum dolor sit amet. Thun amis ozz weepay yusil eepeet zaboy. Istee mundus feeri  
obundus."+HRT+HRT+"Lorem ipsum dolor sit amet istee mundus feeri  
obundus."+HRT+HRT+HRT+Author+HRT+Title

Switch(Format)

CaseOf "Full Block": LetterType:=FullBlock  
CaseOf "Modified Block": LetterType:=ModBlock  
CaseOf "Semi-Block": LetterType:=SemiBlock  
CaseOf "Simplified": LetterType:=Simplified

EndSwitch

DlgWidth:=320

DlgHeight:=172

ButtonWidth:=60

ButtonHPos:=DlgWidth-ButtonWidth-12

DialogDefine("Opening";50;50;DlgWidth;DlgHeight;2+16+512;"Letter")

DialogAddPushButton("Opening";"OK";ButtonHPos;8;ButtonWidth;13;1;"OK")

DialogAddPushButton("Opening";"Cancel";ButtonHPos;24;ButtonWidth;13;0;"Cancel")

DialogAddPushButton("Opening";"AddressBook";ButtonHPos;45;ButtonWidth;13;0;"&Address  
Book...")

DialogAddPushButton("Opening";"PersonalInfo";ButtonHPos;61;ButtonWidth;13;0;"&Personal  
Info...")

```

DialogAddText("Opening";"NameStatic";104;8;110;10;1;"&Recipient's Name and Address:")
DialogAddEditBox("Opening";"RecipAddress";104;19;128;45;1+32+64;LtrAddress;0)
DialogAddText("Opening";"SalutStatic";104;70;125;10;1;"&Salutation (E.g.: Dear Ms.
Smith):")
DialogAddEditBox("Opening";"SalutationBox";104;81;128;13;1+32;Salutation;0)
DialogAddCheckBox("Opening";"Header";114;108;100;11;"&Header for Second
Page";HeaderOn)
DialogAddCheckBox("Opening";"Subject";114;120;100;11;"Su&bje&ct/Reference
Line";SubjectOn)
DialogAddCheckBox("Opening";"Return";114;132;100;11;"Se&nder's Return
Address";ReturnOn)
DialogAddGroupBox("Opening";"GB1";104;98;128;50;"Include")
DialogAddText("Opening"; "FormatStatic";8;8;70;10;1;"Letter &Format:")
DialogAddText("Opening";"LetterFrame";8;20;80;109;1+8;""")
DialogAddText("Opening";"LetterPicture";13;25;70;99;1+16;LetterType)
DialogAddPopUpButton("Opening";"Format";8;135;80;13;Format)
DialogAddListItem("Opening";"Format";"Full Block")
DialogAddListItem("Opening";"Format";"Modified Block")
DialogAddListItem("Opening";"Format";"Semi-Block")
DialogAddListItem("Opening";"Format";"Simplified")
DialogAddVLine("Opening";"VL1";96;0;DlgHeight)
DialogAddVLine("Opening";"VL2";240;0;DlgHeight)
If(Format="Simplified")
    RegionSetWindowText("Opening.SalutStatic";"<No Salutation with Simplified
Format>")
    RegionShowWindow("Opening.SalutationBox";Hide!)
EndIf
RegionShowWindow("Opening.CancelBtt";Hide!)
Return
//*****

//*****
//    ROUTINE NAME: DisplayMainDlg@
//    INPUT VARIABLES: None
//    OUTPUT VARIABLES: None
//    USES: Nothing
//    DESCRIPTION: Displays main dialog and begins callback routine.
//*****
Label(DisplayMainDlg@)
// Set WYSBYGI font
hFont:=SelectFont("Arial";6)
SetCtrlFont("Opening";"LetterPicture";hFont)
Display(On!)
DialogDisplay("Opening";"RecipAddress";Callback@)
NotDone:=True
While(NotDone)

```

```

EndWhile
DialogUndisplay("Opening";"OK")
Display(Off!)
Return

Label(Callback@)
Switch(Callback[3])
  CaseOf "OK":
    RegionGetWindowText(LtrAddress;"Opening.RecipAddress")
    If(LtrAddress="")
      MessageBox(;"WordPerfect for Windows";"Please enter the recipient's
name and address.";OK!|IconExclamation!)
      RegionSetFocus("Opening.RecipAddress")
    Else
      NotDone:=False
    EndIf
  CaseOf "Cancel";"CancelBttn":
    DialogUndisplay("Opening";"Cancel")
    Go(Stop@)
  CaseOf "AddressBook":
    AddressBookDlg ()
    If(?AddressBookSelection<>"")
      AdrsString:=?AddressBookSelection
      LtrAddress:=""
      Salutation:=""
      For(Count; 1; Count<=9; Count+1)
// increased count to 8 for UK market.
        DelimPos:=StrPos(AdrsString; "|")
        NextFld:=SubStr(AdrsString; 1; DelimPos-1)
        If(NextFld<>"")
          If(Count<>2)
            If(LtrAddress<>"")
              LtrAddress:=LtrAddress+ NToC(63754)+
NextFld
            Else
              LtrAddress:=NextFld
            EndIf
          Else
            Salutation:=NextFld
          EndIf
        EndIf
      AdrsString:=SubStr(AdrsString;DelimPos+1;StrLen(AdrsString)-
DelimPos)
    EndFor
    RegionSetWindowText("Opening.RecipAddress";LtrAddress)
    RegionSetWindowText("Opening.SalutationBox";Salutation)

```

```

        RegionSetFocus("Opening.OK")
    Return
Else
    RegionSetFocus("Opening.RecipAddress")
    Return
EndIf
CaseOf "PersonalInfo":
    FillAgain:=True
    PersonalInfoDlg()
    RegionSetFocus("Opening.RecipAddress")
CaseOf "":
    Switch(Callback[6])
        CaseOf 1000:
            RegionSetWindowText("Opening.LetterPicture";FullBlock)
            RegionSetWindowText("Opening.SalutStatic";"&Salutation (E.g.: Dear Ms. Smith):")
            RegionShowWindow("Opening.SalutationBox";Show!)
        CaseOf 1001:
            RegionSetWindowText("Opening.LetterPicture";ModBlock)
            RegionSetWindowText("Opening.SalutStatic";"&Salutation (E.g.: Dear Ms. Smith):")
            RegionShowWindow("Opening.SalutationBox";Show!)
        CaseOf 1002:
            RegionSetWindowText("Opening.LetterPicture";SemiBlock)
            RegionSetWindowText("Opening.SalutStatic";"&Salutation (E.g.: Dear Ms. Smith):")
            RegionShowWindow("Opening.SalutationBox";Show!)
        CaseOf 1003:
            RegionSetWindowText("Opening.LetterPicture";Simplified)
            RegionSetWindowText("Opening.SalutStatic";"<No Salutation with Simplified Format>")
            RegionShowWindow("Opening.SalutationBox";Hide!)
    EndSwitch
EndSwitch
Return
//*****

//*****
//    ROUTINE NAME: SaveSettings@
//    INPUT VARIABLES: Format; SubjectOn; ReturnOn
//    OUTPUT VARIABLES: None
//    USES: Nothing
//    DESCRIPTION: Stores letter settings to private BIF file.
//*****
Label(SaveSettings@)
BIFWrite(Status:BIFStat; Group: GroupName; Section: "Letters"; Item: "Format"; Value:

```

```

Format; ItemType: WPWordString!)
BIFWrite(Status:BIFStat; Group: GroupName; Section: "Letters"; Item: "Print Subject"; Value:
SubjectOn; ItemType: Boolean!)
BIFWrite(Status:BIFStat; Group: GroupName; Section: "Letters"; Item: "Print Return Address";
Value: ReturnOn; ItemType: Boolean!)
BIFWrite(Status:BIFStat; Group: GroupName; Section: "Letters"; Item: "Header for Second
Page"; Value: HeaderOn; ItemType: Boolean!)
Return
//*****

//*****
//      ROUTINE NAME: DoLetter@
//      INPUT VARIABLES: None
//      OUTPUT VARIABLES: None
//      USES: Nothing
//      DESCRIPTION: Builds letter.
//*****
Label(DoLetter@)
If (FillAgain)
    TemplateFill ()
EndIf
OnError(NoBookmark@)
PosDocTop()
SelectMode(On!)
BookmarkFind("Letter Text")
DeleteCharNext()
Label(NoBookmark@)
OnError(Error@)
If(?BlockActive<>0)
    SelectMode(Off!)
EndIf
If(HeaderOn)
    MacroStatusPrompt(On!; "Creating header for second page...")
    HeaderB (Create!)
    // Find end of first line of address
    HrtPos:=CharPos(LtrAddress;NToC(63754))
    If(HrtPos=0)
        Type(LtrAddress)
    Else
        HrtPos:=HrtPos-1
        Type(SubChar(LtrAddress;1;HrtPos))
    EndIf
    If((Format="Modified Block") Or (Format="Semi-Block"))
        Center() Type("- ") PageNumberDisplay() Type(" -")
        FlushRight()
        DateText ()

```

```

Else
    HardReturn ()
    Type ("Page ")
    PageNumberDisplay()
    HardReturn()
    DateText()
EndIf
NewHFWOccurrence (AllPages!)
SubstructureExit ()
Suppress (HeaderB!)
MacroStatusPrompt (Off!)
Else
    If(?HeaderBOccur<>0)
        HeaderB(Off!)
    EndIf
EndIf
OnError Call (CreateStyle@)
StyleEditBegin ("Closing"; CurrentDoc!)
StyleCodes (WithoutOffCodes!; CurrentDoc!)
OnError (Error@)
If((Format="Modified Block") Or (Format="Semi-Block"))
    CenterPos:=Integer((?PaperWidth-?MarginLeft-?MarginRight)/2)
    FirstLineIndent(0)
    ParagraphSpacing(1)
    MarginAdjustLeft(CenterPos)
Else
    CenterPos:=0
    FirstLineIndent(0)
    ParagraphSpacing(1)
    MarginAdjustLeft (0)
EndIf
SubstructureExit ()
StyleEditEnd (Save!)
MarginAdjustLeft(CenterPos)
FirstLineIndent(0)
ParagraphSpacing(1)
LineSpacing(1)
If(ReturnOn)
    // Retrieve personal information from private BIF.
    ReturnInfo:=""
    PersInfo:=""
    CurrField:=""
    BIFRead(Status:BIFStat; Group:GroupName; Section: "Template Information"; Item:
    "Personal Information"; Value: PersInfo; ExpectedItemType: WPWordString!;
    NameListType: Private!)
    If(PersInfo<> "")

```

```

For(Count;1;Count<=7;Count+1)
// changed count to 7 for UK market
  CurrField:=SubStr(PersInfo;1;StrPos(PersInfo;"|")-1)
  If(CurrField<>"")
    ReturnInfo:=ReturnInfo+CurrField+NToC(63754)
  EndIf
  PersInfo:=SubStr(PersInfo;StrPos(PersInfo;"|")+1;StrLen(PersInfo))
EndFor
Type(ReturnInfo)
HardReturn()
EndIf
Discard(ReturnInfo; PersInfo; CurrField)
EndIf
DateText()
HardReturn()
MarginAdjustLeft (0)
HardReturn()
HardReturn()
HardReturn()
Type(LtrAddress)
HardReturn()
HardReturn()
If(SubjectOn)
  DlgWidth:=222
  DlgHeight:=71
  ButtonWidth:=35
  ButtonHPos:=DlgWidth-ButtonWidth-12
  DialogDefine("Subject";50;50;DlgWidth;DlgHeight;16;"Subject / Reference Line")
  DialogAddPushButton("Subject";"OK";94;37;ButtonWidth;13;1;"OK")
  DialogAddText("Subject";"T1";8;8;100;10;1;"Enter the subject of this letter:")
  DialogAddEditBox("Subject";"Subject";8;19;200;13;1;Subject;0)
  Display(On!)
  DialogDisplay("Subject";"Subject")
  Display(Off!)
  DialogDestroy("Subject")
  AttributeAppearanceOn(Bold!)
  Type(SubjectPrefix+Subject+" ")
  AttributeAppearanceOff(Bold!)
  HardReturn()
  HardReturn()
EndIf
If((Format<>"Simplified") And (Salutation<>""))
  LastChar:=SubStr(Salutation;StrLen(Salutation);1)
  If((LastChar=":") Or (LastChar=",")) Or (LastChar="-"))
    Type(Salutation)
  Else

```

```

        Type(Salutation+":")
    EndIf
    HardReturn()
    HardReturn()
EndIf
ParagraphSpacing(1.75)
If(Format="Semi-Block")
    FirstLineIndent(0.375")
EndIf
BookmarkCreate("Letter Text")
If(Exists(LetterExpert)<>3)
    If(ToUpper(?CurrentTemplate)<>ToUpper(?TemplateFile))
        Display(On!)
        MessageBox("Letter Formatting Complete";"To insert the letter closing, choose
"+HRT+"""Letter Closing"" from the ""Insert"" menu.";OK!|IconInformation!)
    EndIf
EndIf
Return
//*****

Label (CreateStyle@)
StyleCreate ("Closing"; OpenStyle!; ; CurrentDoc!)
StyleEditBegin ("Closing"; CurrentDoc!)
StyleDescription ("Letter Closing")
Return

Label(Error@)
MessageBox("Unexpected Error";"An unexpected error has occurred.";OK!|IconStop!)
Quit

// End of macro
Label(Stop@)
DialogDestroy("Opening")
FreeFont(hFont)
Quit

//*****
//      FUNCTION: SelectFont(FontName, FontSize)
//      RETURN: Handle to font
//*****
FUNCTION SelectFont( FontName; FontSize)
GLink := DllLoad("GDI")
DllCall(GLink;"CreateFont"; hFont;WORD;{
LOWORD(FontSize);
LOWORD(0);LOWORD(0);LOWORD(0);LOWORD(0);
LOWORD(0);LOWORD(0);LOWORD(0);LOWORD(0);

```

```

LOWORD(0);LOWORD(0);LOWORD(0);LOWORD(0);
FontName
})
DLLFree(GLink)
RETURN (hFont)
ENDFUNC
//*****

//*****
//      PROCEDURE: SetCtrlFont( DlgID, CtrlID, hFont)
//*****
PROCEDURE SetCtrlFont( DlgID; CtrlID; hFont)
ULink := DLLLoad ("USER")
DialogHandle(hwnd; DlgID; CtrlID)
DLLCall(ULink;"SendMessage";ret;DWORD;{
LOWORD(hwnd);
LOWORD(48); //WM_SETFONT
LOWORD(hFont);
1
})
DLLFree (ULink)
ENDPROC
//*****

//*****
//      PROCEDURE: FreeFont( hFont)
//*****
PROCEDURE FreeFont( hFont)
GLink := DLLLoad ("GDI")
DLLCall(GLink;"DeleteObject";Ret;WORD;{LOWORD(hFont)})
DLLFree (GLink)
ENDPROC
//*****

```